Advanced Engineering Mathematics 1. 2-Hours Exam May 14, 2018.

JE 9.5.18 JKL 19.5.18

Problem 1

```
> restart:with(LinearAlgebra):with(plots):
```

A real function f of two real variables is given by

```
> f:=(x,y)->4*y*(x^2+1/3*y^2-1);
```

$$f := (x, y) \rightarrow 4\, y \left(x^2 + \frac{1}{3}\, y^2 - 1\right) \tag{1.1}$$

```
> expand(f(x,y));
```

$$4\, y\, x^2 + \frac{4}{3}\, y^3 - 4\, y \tag{1.2}$$

Question 1

If f has a local extremum in a point, then the point must be a stationary point since f has no exception points.

$\nabla f(x, y) = \left(f'_x(x, y), f'_y(x, y)\right) = (8\,xy, 4\,x^2 + 4\,y^2 - 4) = (0, 0) \Leftrightarrow$

$8xy = 0$ and $4x^2 + 4\,y^2 - 4 = 0 \Leftrightarrow x = 0$ and $4y^2 - 4 = 0$ or $y = 0$ and $4x^2 - 4 = 0 \Leftrightarrow x = 0$ and $y = \pm 1$ or $y = 0$ and $x = \pm 1$.

All stationary points for f are then $(1, 0)$, $(-1, 0)$, $(0, 1)$ and $(0, -1)$.

The Hessian matrix for f in the point (x, y) is

```
> H(x,y):=<diff(f(x,y),x,x),diff(f(x,y),y,x)|diff(f(x,y),x,y),
  diff(f(x,y),y,y)>;
```

$$H(x, y) := \begin{bmatrix} 8\,y & 8\,x \\ 8\,x & 8\,y \end{bmatrix} \tag{1.1.1}$$

```
> H(1,0):=subs(x=1,y=0,H(x,y));
```

$$H(1, 0) := \begin{bmatrix} 0 & 8 \\ 8 & 0 \end{bmatrix} \tag{1.1.2}$$

```
> Eigenvalues(H(1,0),output=list);
```

$$[8, -8] \tag{1.1.3}$$

Since the two eigenvalues for **H**(1,0) have opposite signs, f does not have a local extremum in the stationary point (1,0) (saddle point).

```
> H(-1,0):=subs(x=-1,y=0,H(x,y));
```

$$H(-1, 0) := \begin{bmatrix} 0 & -8 \\ -8 & 0 \end{bmatrix} \tag{1.1.4}$$

```
> Eigenvalues(H(-1,0),output=list);
```

$$[8, -8] \tag{1.1.5}$$

Since the two eigenvalues for **H**(-1,0) have opposite signs, f does not have a local extremum in the stationary point (1,0) (saddle point).

```
> H(0,1):=subs(x=0,y=1,H(x,y));
```

$$H(0,1) := \begin{bmatrix} 8 & 0 \\ 0 & 8 \end{bmatrix} \tag{1.1.6}$$

```
> Eigenvalues(H(0,1),output=list);
```

$$[8, 8] \tag{1.1.7}$$

Since both eigenvalues for **H**(0,1) are positive, f has a proper local minimum in the stationary point (0, 1) with the value

```
> 'f(0,1)'=f(0,1);
```

$$f(0,1) = -\frac{8}{3} \tag{1.1.8}$$

```
> H(0,-1):=subs(x=0,y=-1,H(x,y));
```

$$H(0,-1) := \begin{bmatrix} -8 & 0 \\ 0 & -8 \end{bmatrix} \tag{1.1.9}$$

```
> Eigenvalues(H(0,-1),output=list);
```

$$[-8, -8] \tag{1.1.10}$$

Since both the eigenvalues for **H**(0, −1) are negative, f has a proper local maximum in the stationary point(0, −1) with the value

```
> 'f(0,-1)'=f(0,-1);
```

$$f(0,-1) = \frac{8}{3} \tag{1.1.11}$$

```
> contourplot(f(x,y),x=-3..3,y=-3..3,contours=80);
```



An ellipse E in the (x, y)-plane is given by the equation $x^2 + \frac{y^2}{3} - 1 = 0 \Leftrightarrow \frac{x^2}{1} + \frac{y^2}{3} = 1$.

Question 2

The ellipse E has the centre in $(0, 0)$, the semi major axis $a = \sqrt{3}$ and semi minor axis $b = 1$.

Question 3

We consider the bounded and closed set of points M bounded by E.

```
> implicitplot(x^2+y^2/3=1,x=-2..2,y=-2..2,scaling=constrained,
  filled=true,coloring=[gray,white], title="M");
```



Since M is bounded and closed and since f is continuous in M, f has a global minimum and a global maximum in M. Since f has no exception points in the interiour of M, these values are attained either in a stationary point in the interiour of M or on the boundary of M.

The only stationary points in the interiour of M are $(0, 1)$ and $(0, -1)$ and

```
> 'f(0,1)'=f(0,1);'f(0,-1)'=f(0,-1);
```

$$f(0, 1) = -\frac{8}{3}$$

$$f(0, -1) = \frac{8}{3} \quad \textbf{(1.3.1)}$$

Since the boundary E af M is given by the equation $x^2 + \frac{y^2}{3} - 1 = 0$, $f(x, y) = 0$ for all $(x, y) \in E$.

$$x^2 + \frac{1}{3}y^2 - 1 = 0 \qquad \textbf{(1.3.2)}$$

By numerical comparison of these investigations we find that the global minimum is $-\frac{8}{3}$, that is attained in the point (0, 1) and that the global maximum is $\frac{8}{3}$, that is attained in the point (0, −1).

Note that, since M is conneceted, the range is $f(M) = [-\frac{8}{3}; \frac{8}{3}]$.

```
> plot3d(f(x,y),x=-2..2,y=-2..2,view=-3..3);
```



Problem 2

```
> restart:with(plots):
  prik:=(x,y)->VectorCalculus[DotProduct](x,y):
  kryds:=(x,y)->convert(VectorCalculus[CrossProduct](x,y),Vector)
  :
  vop:=proc(X) op(convert(X,list)) end proc:
```

In the (x, y)-plane we consider the set of points $B = \{ (x, y) \mid -1 \le x \le 1 \text{ and } 0 \le y \le \cosh(x) \}$.

```
> plot(cosh(x),x=-1..1,scaling=constrained,filled=true,title="M")
;
```



Question 1

A parametric representation for B is

$$\mathbf{r}(u, v) = \begin{bmatrix} u \\ v\cosh(u) \end{bmatrix}, u \in [-1;1], v \in [0;1].$$

A height function h defined on the (x, y)-plane is given by $z = h(x, y)$, where

```
> h:=(x,y)->2-x;
```

$$h := (x, y) \to 2 - x \tag{2.1}$$

Let F denote the part of the graph for h that lies vertically above B.

Question 2

A parametric representation for F is

```
> r:=<u,v*cosh(u),h(u,v*cosh(u))>;
```

(2.2.1)

$$r := \begin{bmatrix} u \\ v\cosh(u) \\ 2-u \end{bmatrix} \quad (2.2.1)$$

where $u \in [-1;1]$ and $v \in [0;1]$.

```
> ru:=diff~(r,u);
```

$$ru := \begin{bmatrix} 1 \\ v\sinh(u) \\ -1 \end{bmatrix} \quad (2.2.2)$$

```
> rv:=diff~(r,v);
```

$$rv := \begin{bmatrix} 0 \\ \cosh(u) \\ 0 \end{bmatrix} \quad (2.2.3)$$

The normal vector of the surface is

```
> N:=kryds(ru,rv);
```

$$N := \begin{bmatrix} \cosh(u) \\ 0 \\ \cosh(u) \end{bmatrix} \quad (2.2.4)$$

The Jacobi function corresponding to **r** is

```
> Jacobi:=sqrt(prik(N,N)) assuming cosh(u)>0;
```

$$Jacobi := \sqrt{2}\,\cosh(u) \quad (2.2.5)$$

Question 3

```
> f:=(x,y,z)->(z-1)/sqrt(2);
```

$$f := (x, y, z) \rightarrow \frac{z-1}{\sqrt{2}} \quad (2.3.1)$$

$$\int_F \frac{(z-1)}{\sqrt{2}}\,\mathrm{d}\mu = \int_{v=0}^{1}\int_{u=-1}^{1} \frac{(z(u,v)-1)}{\sqrt{2}}\,\mathrm{Jacobi}(u,v)\,\mathrm{d}u\,\mathrm{d}v =$$

$$\int_{v=0}^{1}\int_{u=-1}^{1} (1-u)\cosh(u)\,\mathrm{d}u\,\mathrm{d}v\,.$$

```
> integranden:=f(vop(r))*Jacobi;
```

$$integranden := (1-u)\cosh(u) \quad (2.3.2)$$

```
> Int(Int(integranden,u=-1..1),v=0..1)=int(int(integranden,u=
  -1..1),v=0..1);
```

$$\int_0^1\int_{-1}^1 (1-u)\cosh(u)\,\mathrm{d}u\,\mathrm{d}v = 2\sinh(1) \quad (2.3.3)$$

Problem 3

```
> restart:with(LinearAlgebra):with(plots):
  vop:=proc(X) op(convert(X,list)) end proc:
```

It is given that

$$\begin{bmatrix} x'(t) \\ y'(t) \end{bmatrix} = \begin{bmatrix} 4\,x(t) - 10\,y(t) \\ 2\,x(t) - 5\,y(t) \end{bmatrix}, t \in \mathbb{R},$$

has the complete solution

$$\begin{bmatrix} x(t) \\ y(t) \end{bmatrix} = \begin{bmatrix} 5\,c_1 + 2\,c_2 e^{-t} \\ 2\,c_1 + c_2 e^{-t} \end{bmatrix}, t \in \mathbb{R}, c_1, c_2 \in \mathbb{R},$$

In the (x, y)-plane we consider the vector field $\mathbf{V}(x, y) = \begin{bmatrix} 4\,x - 10\,y \\ 2\,x - 5\,y \end{bmatrix}$.

Question 1

The complete solution given above is precisely the flow curves for the plane vector field $\mathbf{V}(x, y) = \begin{bmatrix} 4\,x - 10\,y \\ 2\,x - 5\,y \end{bmatrix}$.

To determine the constants c_1 and c_2 such that $\begin{bmatrix} x(0) \\ x(0) \end{bmatrix} = \begin{bmatrix} 1 \\ 1 \end{bmatrix}$ we have the system of linear equations

$$\begin{bmatrix} x(0) \\ y(0) \end{bmatrix} = \begin{bmatrix} 5\,c_1 + 2\,c_2 \\ 2\,c_1 + c_2 \end{bmatrix} = \begin{bmatrix} 5 & 2 \\ 2 & 1 \end{bmatrix} \begin{bmatrix} c_1 \\ c_2 \end{bmatrix} = \begin{bmatrix} 1 \\ 1 \end{bmatrix}$$

with the corresponding augmented matrix

```
> T:=<<5,2>|<2,1>|<1,1>>;
```

$$T := \begin{bmatrix} 5 & 2 & 1 \\ 2 & 1 & 1 \end{bmatrix} \qquad \textbf{(3.1.1)}$$

```
> c:=LinearSolve(T);
```

$$c := \begin{bmatrix} -1 \\ 3 \end{bmatrix} \qquad \textbf{(3.1.2)}$$

The wanted flow curve is then

$$\begin{bmatrix} x(t) \\ y(t) \end{bmatrix} = \begin{bmatrix} -5 + 6e^{-t} \\ -2 + 3e^{-t} \end{bmatrix}, t \in \mathbb{R}.$$

L is the straight line segment from the point (1, 1) to the point (2, 2).

Question 2

A parametric representaion for L is

$$\begin{bmatrix} x(u) \\ y(u) \end{bmatrix} = \begin{bmatrix} u \\ u \end{bmatrix}, \text{ u} \in [1;2].$$

Question 3

To determine the constants c_1 and c_2 such that $\begin{bmatrix} x(0) \\ x(0) \end{bmatrix} = \begin{bmatrix} u \\ u \end{bmatrix}$ we have the system of linear equations

$$\begin{bmatrix} x(0) \\ y(0) \end{bmatrix} = \begin{bmatrix} 5\,c_1 + 2\,c_2 \\ 2\,c_1 + c_2 \end{bmatrix} = \begin{bmatrix} 5 & 2 \\ 2 & 1 \end{bmatrix} \begin{bmatrix} c_1 \\ c_2 \end{bmatrix} = \begin{bmatrix} u \\ u \end{bmatrix}$$

with the corresponding augmented matrix

$$-1$$

$$3$$

$$\begin{bmatrix} x(0) \\ x(0) \end{bmatrix}$$

$$\begin{bmatrix} u \\ u \end{bmatrix} \tag{3.3.1}$$

```
> T:=<<5,2>|<2,1>|<u,u>>;
```

$$T := \begin{bmatrix} 5 & 2 & u \\ 2 & 1 & u \end{bmatrix} \tag{3.3.2}$$

```
> c:=LinearSolve(T);
```

$$c := \begin{bmatrix} -u \\ 3\,u \end{bmatrix} \tag{3.3.3}$$

The flow curve **r**(u, t) for **V**, that fulfills **r**$(u, 0) = \begin{bmatrix} u \\ u \end{bmatrix}$ then has the parametric representation

```
> r:=(u,t)-><-5*u+6*u*exp(-t),-2*u+3*u*exp(-t)>:
> r(u,t);
```

$$\begin{bmatrix} -5\,u + 6\,u\,e^{-t} \\ -2\,u + 3\,u\,e^{-t} \end{bmatrix} \tag{3.3.4}$$

For $t = 1$ and $u \in [1; 2]$ we get

```
> r(u,1);
```

$$\begin{bmatrix} -5\,u + 6\,u\,e^{-1} \\ -2\,u + 3\,u\,e^{-1} \end{bmatrix} \tag{3.3.5}$$

that is a parametric representaion of the curve that L has been deformed to, at time t = 1.

Since **r**$(u, 1)$ can be written as **r**$(u, 1) = u \begin{bmatrix} -5 + 6e^{-1} \\ -2 + 3e^{-1} \end{bmatrix}$, where $u \in [1; 2]$, the curve is the straight line segment from the point $(-5 + 6e^{-1}, -2 + 3e^{-1})$ to the point $(-10 + 12e^{-1}, -4 + 6e^{-1})$.

```
> L:=plot(x,x=1..2,thickness=5,color=blue):
> P1:=plot([vop(r(u,1)),u=1..2],thickness=5,color=red):
> K1:=plot([vop(r(1,t)),t=0..1],linestyle=4):
> K2:=plot([vop(r(2,t)),t=0..1],linestyle=4):
> display(L,P1,K1,K2,scaling=constrained);
```



Problem 4

```
> restart:with(plots):
  prik:=(x,y)->VectorCalculus[DotProduct](x,y):
  kryds:=(x,y)->convert(VectorCalculus[CrossProduct](x,y),Vector)
  :
  vop:=proc(X) op(convert(X,list)) end proc:
  grad:=X->convert(Student[VectorCalculus][Del](X),Vector):
  div:=V->VectorCalculus[Divergence](V):
  rot:=proc(X) uses VectorCalculus;BasisFormat(false);Curl(X)
  end proc:
```

In the (x, y, z)-space we are given the vector field

```
> V:=(x,y,z)-><3*x-y^2,4*y*z,y-2*z^2>:
> V(x,y,z);
```

$$\begin{bmatrix} -y^2 + 3x \\ 4yz \\ -2z^2 + y \end{bmatrix} \quad \textbf{(4.1)}$$

Ω is a solid cylinder of revolution with radius 1, the x-axis as the axis of symmetry and bounded bounded by the planes $x = 0$ and $x = 1$ (see the figure in the text of the prolem).

Question 1

$\partial\Omega$ is the closed surface of Ω with an orientation given by an outward pointing unit normal vector.
From Gauss' Theorem we get

```
> div(V)(x,y,z);
```

$$3 \tag{4.1.1}$$

$$\mathrm{Flux}(\mathbf{V}, \partial\Omega) = \int_{\Omega} \mathrm{Div}(\mathbf{V})\, d\mu = \int_{\Omega} 3\, d\mu = 3\mathrm{Vol}(\Omega) = 3\pi.$$

The circular disc C in the plane $x = 1$ ends the cylinder in the direction of the x-axis and its boundary curve ∂C is shown in red in the figure in he text of the problem.

Question 2

The parametric representation for C

```
> r:=<1,u*cos(v),u*sin(v)>;
```

$$r := \begin{bmatrix} 1 \\ u\cos(v) \\ u\sin(v) \end{bmatrix} \tag{4.2.1}$$

where u $\in$ [0;1] and v $\in$ [0;2π].

```
> ru:=diff~(r,u);
```

$$ru := \begin{bmatrix} 0 \\ \cos(v) \\ \sin(v) \end{bmatrix} \tag{4.2.2}$$

```
> rv:=diff~(r,v);
```

$$rv := \begin{bmatrix} 0 \\ -u\sin(v) \\ u\cos(v) \end{bmatrix} \tag{4.2.3}$$

The normal vector of the surface is

```
> N:=simplify(kryds(ru,rv));
```

$$N := \begin{bmatrix} u \\ 0 \\ 0 \end{bmatrix} \tag{4.2.4}$$

where $u \in \,]0;1]$, is precisely forming a right-hand screw with the chosen orientation of the closed boundary curve ∂C for C (shown with blue tangent vector on the figure in the text of the problem).
Therefore the stated parametric representation for C fulfills the condition.

Question 3

From Stokes' Theorem we then get

$$\text{Cirk}(\mathbf{V},\partial C) = \int_{\partial C} \mathbf{V} \cdot \mathbf{e}_{\partial C}\, d\mu = \text{Flux}(\mathbf{Curl}(\mathbf{V}),C) = \int_{C} \mathbf{n}_C \cdot \mathbf{Curl}(\mathbf{V})\, d\mu =$$

$$\int_{v=0}^{2\pi} \int_{u=0}^{1} \mathbf{N}(u, v) . \mathbf{Curl}(\mathbf{V})(\mathbf{r}(u, v))\, du\, dv .$$

```
> rot(V)(x,y,z);
```

$$\begin{bmatrix} 1 - 4y \\ 0 \\ 2y \end{bmatrix} \qquad (4.3.1)$$

The **Curl** restricted to the surface is

```
> Curl:=rot(V)(vop(r));
```

$$Curl := \begin{bmatrix} 1 - 4u\cos(v) \\ 0 \\ 2u\cos(v) \end{bmatrix} \qquad (4.3.2)$$

```
> integranden:=prik(N,Curl);
```

$$integranden := u\,(1 - 4u\cos(v)) \qquad (4.3.3)$$

```
> Int(Int(integranden,u=0..1),v=0..2*Pi)=int(int(integranden,u=
  0..1),v=0..2*Pi);
```

$$\int_0^{2\pi} \int_0^1 u\,(1 - 4u\cos(v))\, du\, dv = \pi \qquad (4.3.4)$$